

```
> restart;
```

```
> fun := hypergeom([1/4,5/4],[2],1-v^2)^2;
```

$$fun := \text{hypergeom}\left(\left[\frac{1}{4}, \frac{5}{4}\right], [2], -v^2 + 1\right)^2 \quad (1)$$

```
> deq := gfun:-holexprtodiffeq(fun, y(v));
```

$$deq := 15 y(v) v + (33 v^2 - 9) \left(\frac{d}{dv} y(v)\right) + (12 v^3 - 12 v) \left(\frac{d^2}{dv^2} y(v)\right) + (v^4 - 2 v^2 + 1) \left(\frac{d^3}{dv^3} y(v)\right) \quad (2)$$

```
> deq2 := collect(subs(v=u, eval(subs(y(v)=diff(y(v),v), deq))), diff, factor);
```

$$deq2 := (u - 1)^2 (u + 1)^2 \left(\frac{d^4}{du^4} y(u)\right) + 12 u (u - 1) (u + 1) \left(\frac{d^3}{du^3} y(u)\right) + 15 \left(\frac{d}{du} y(u)\right) u + (33 u^2 - 9) \left(\frac{d^2}{du^2} y(u)\right) \quad (3)$$

```
> ini := {y(0)=0, seq((D@@(i+1))(y)(1/2)=simplify(eval(diff(fun, [v$ i]), v=1/2)), i=0..2)};
```

$$ini := \left\{ y(0) = 0, D(y) \left(\frac{1}{2}\right) = \frac{256}{81} \frac{\left(3 \text{EllipticK}\left(\frac{1}{2}\right) - 2 \text{EllipticE}\left(\frac{1}{2}\right)\right)^2}{\pi^2}, D^{(2)}(y) \left(\frac{1}{2}\right) \right. \\ = -\frac{5}{16} \text{hypergeom}\left(\left[\frac{1}{4}, \frac{5}{4}\right], [2], \frac{3}{4}\right) \text{hypergeom}\left(\left[\frac{5}{4}, \frac{9}{4}\right], [3], \frac{3}{4}\right), D^{(3)}(y) \left(\frac{1}{2}\right) \\ = \frac{29}{4} \text{hypergeom}\left(\left[\frac{1}{4}, \frac{1}{4}\right], [3], \frac{3}{4}\right)^2 - \frac{299}{64} \text{hypergeom}\left(\left[\frac{1}{4}, \frac{1}{4}\right], [3], \frac{3}{4}\right) \text{hypergeom}\left(\left[\frac{5}{4}, \frac{5}{4}\right], [4], \frac{3}{4}\right) + \frac{375}{512} \text{hypergeom}\left(\left[\frac{5}{4}, \frac{5}{4}\right], [4], \frac{3}{4}\right)^2 \left. \right\} \quad (4)$$

```
> sol := (rhs(dsolve({deq2} union ini))):
```

```
> fsol := evalf(sol);
```

$$fsol := (2.666666444 - 2.050497521 \cdot 10^{-7} I) \text{hypergeom}([0.5000000000, 0.5000000000, 2.5000000000], [1., 2.], -1. (u - 1.) (u + 1.)) u + (-7.439281853 \cdot 10^{-9} + 5.720336350 \cdot 10^{-16} I) \text{MeijerG}([[-1.5000000000, 0.5000000000, 0.5000000000], [ ]], [[0., 0., -1.], [ ]], (u - 1.) (u + 1.)) u + \frac{1}{u^2 - 1} ((-7.800458658 \cdot 10^{-16} - 1.014447525 \cdot 10^{-8} I) u \text{MeijerG}([[-0.5000000000, 1.5000000000, 1.5000000000], [ ]], [[1., 1.], [0.]], -1. (u - 1.) (u + 1.))) \quad (5)$$

```
> mysol := (8/3)*u*hypergeom([1/2,1/2,5/2],[1,2],1-u^2);
```

$$mysol := \frac{8}{3} \text{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}, \frac{5}{2}\right], [1, 2], -u^2 + 1\right) u \quad (6)$$

```
> ref:=Int(fun,v=0..u);
```

(7)

$$ref := \int_0^u \text{hypergeom}\left(\left[\frac{1}{4}, \frac{5}{4}\right], [2], -v^2 + 1\right)^2 dv \quad (7)$$

> identify(evalf[30](subs(u=1/4, mysol-ref)));

$$\frac{32}{9\pi} \quad (8)$$

> mysol := -32/(9\*Pi) + (8/3)\*u\*hypergeom([1/2,1/2,5/2],[1,2],1-u^2);

$$mysol := -\frac{32}{9\pi} + \frac{8}{3} \text{hypergeom}\left(\left[\frac{1}{2}, \frac{1}{2}, \frac{5}{2}\right], [1, 2], -u^2 + 1\right) u \quad (9)$$

> evalf(subs(u=1/4, mysol-ref))

$$4.10^{-10} \quad (10)$$

> evalf(subs(u=17/53, mysol-ref));

$$1.10^{-10} \quad (11)$$

> simplify(eval(deq2,y=unapply(mysol,u)) );

$$0 \quad (12)$$

> MultiSeries:-series(mysol, u=1);

$$\frac{8}{9} \frac{-4 + 3\pi}{\pi} + u - 1 - \frac{5}{16} (u - 1)^2 + \frac{95}{768} (u - 1)^3 - \frac{875}{16384} (u - 1)^4 + \frac{3157}{131072} (u - 1)^5 + O((u - 1)^6) \quad (13)$$

> MultiSeries:-series(ref, u=1);

$$\int_0^1 \text{hypergeom}\left(\left[\frac{1}{4}, \frac{5}{4}\right], [2], -v^2 + 1\right)^2 dv + u - 1 - \frac{5}{16} (u - 1)^2 + \frac{95}{768} (u - 1)^3 - \frac{875}{16384} (u - 1)^4 + \frac{3157}{131072} (u - 1)^5 + O((u - 1)^6) \quad (14)$$